# sh testcase for sts.l pr, @-$rn -*- Asm -*-
# mach: all
# as: -isa=shcompact
# ld: -m shelf32

	.include "compact/testutils.inc"

	start

	.global stsl_pr
stsl_pr:
	# Build up a distinctive bit pattern.
	mov #1, r0
	shll8 r0
	add #12, r0
	shll8 r0
	add #85, r0
	shll8 r0
	add #170, r0
	
	lds r0, pr
	mov #40, r2
	shll8 r2
	# Preserve r2.
	mov r2, r7
	sts.l pr, @-r2

	# check results.
	mov.l @r2, r3
	cmp/eq r0, r3
	bf wrong

	# Ensure decrement occurred.
	add #4, r2
	cmp/eq r2, r7
	bf wrong

okay:
	pass
wrong:
	fail
	
